# Maintainer: Muflone http://www.muflone.com/contacts/english/
# Contributor: yugrotavele <yugrotavele at archlinux dot us>

pkgname=gmtp
pkgver=1.3.11
pkgrel=3
pkgdesc="A simple MP3 player client for MTP based devices"
arch=('loong64' 'x86_64')
url="http://gmtp.sourceforge.net/"
license=('BSD')
depends=('flac' 'gtk3' 'libmtp' 'libid3tag' 'libvorbis')
source=("${pkgname}-${pkgver}.tar.gz"::"https://sourceforge.net/projects/gmtp/files/gMTP-${pkgver}/${pkgname}-${pkgver}.tar.gz/download")
sha256sums=('5a830c9c448aa2caf7a844aaa9ba38c2ac0239547366c1e6874e0080bd5f0613')

build() {
  cd "${pkgname}-${pkgver}"
  CFLAGS+=' -fcommon' ./configure --prefix=/usr
  make
}

package() {
  cd "${pkgname}-${pkgver}"
  install -D -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
  make DESTDIR="${pkgdir}" install
}
